

9/8.8.1

Initialisation d'un mode d'impression pour DMP 2000

Dans ce chapitre, nous vous proposons une RSX bien pratique qui vous permettra d'initialiser votre imprimante DMP 2000 ou compatible EPSON dans l'un des modes d'impression suivants :

- mini ;
- proportionnel ;
- condensé ;
- standard NLQ ;
- proportionnel NLQ ;
- double frappe ;
- italique ;
- gras ;
- souligné ;
- double largeur.

L'intérêt de cette RSX réside dans le fait qu'elle peut cohabiter en mémoire avec un autre programme, Basic ou Assembleur. Elle peut même être activée par ce programme à l'aide de l'instruction !IMP.

COMMENT UTILISER LE PROGRAMME

Si vous désirez utiliser le programme sous sa forme Assembleur, saisissez le listing de la page suivante :

```

1          ORG 9000H
2          LOAD 9000H
3          ;
4          ;-----
5          ; RSX Initialisation d'un mode
6          ; d'impression pour DMP 2000
7          ;-----
8          ;
9 9000 C36D91          JP  DEFRSX          ;Definition RSX
10         ;
11         ;-----
12         ; Declaration des constantes et
13         ; variables du programme
14         ;-----
15         ;
16         SAVA:      DS  1          ;Sauvegarde de A
17 9004 00000000 EVBL:  DB  0,0,0,0
18 9008 00000000          DB  0,0,0,0
19 900C 00000000          DB  0,0,0,0,0
19 9010 00
20         HE:..      DS  1          ;Champ Heures
21         MI:..      DS  1          ;Champ Minutes
22         SE:..      DS  1          ;Champ Secondes
23         BUF:..     DS  4          ;Zone RAM pour LOG EXT
24 9018 1D90 PTRTAB:  DW  TABLE    ;Pointeur TABLE
25 901A C37791          JP  TRAITE    ;Traitement
26 901D 494D TABLE:  DB  "IM"
27 901F D0             DB  "P"+80H
28 9020 00             DB  0          ;Fin de table
29         ;
30         ; Table des modes

```

```

31          ;
32          ONOFF:      EQU  $
33 9021 00000000      DB  0,0,0,0,0,0,0,0,0,0,
33 9025 00000000
33 9029 0000
34          ;
35          CONTROL:   EQU  $
36 902B 1B4DFF1B      DB  27,"M",T,27,"P",T,
36 902F 50FF1B50
36 9033 01FF
37 9035 1B5000FF      DB  27,"P",0,T,15,T,18
37 9039 0FFF12FF
37 903D 1B7801FF
38 9041 1B7800FF      DB  27,"x",0,T,27,"x",
38 9045 1B78011B
38 9049 7001FF
39 904C 1B78001B      DB  27,"x",0,27,"p",0,
39 9050 7000FF1B
39 9054 47FF
40 9056 1B48FF1B      DB  27,"H",T,27,"4",T,
40 905A 34FF1B35
40 905E FF
41 905F 1B45FF1B      DB  27,"E",T,27,"F",T,
41 9063 46FF1B2D
41 9067 01FF
42 9069 1B2D00FF      DB  27,"-",0,T,27,"W",
42 906D 1B5701FF
42 9071 1B57
43 9073 00FF          DB  0,T

```

```
44      ;
45      ;Table des adresses dans CONTROL
46      ;
47      ADR:      EQU $
48 9075 2B90      DW CONTROL
49 9077 2E90      DW CONTROL+3
50 9079 3190      DW CONTROL+6
51 907B 3590      DW CONTROL+10
52 907D 3990      DW CONTROL+14
53 907F 3B90      DW CONTROL+16
54 9081 3D90      DW CONTROL+18
55 9083 4190      DW CONTROL+22
56 9085 4590      DW CONTROL+26
57 9087 4C90      DW CONTROL+33
58 9089 5390      DW CONTROL+40
59 908B 5690      DW CONTROL+43
60 908D 5990      DW CONTROL+46
61 908F 5C90      DW CONTROL+49
62 9091 5F90      DW CONTROL+52
63 9093 6290      DW CONTROL+55
64 9095 6590      DW CONTROL+58
65 9097 6990      DW CONTROL+62
66 9099 6D90      DW CONTROL+66
67 909B 7190      DW CONTROL+70
68      ;
69      ; Explicitation des modes
70      ;
71      AUSEC:    EQU $
72 909D 20312920  DB " 1) Caract mini",
72 90A1 43617261
72 90A5 6374206D
```

72 90A9 696E690D	
72 90AD 0A	
73 90AE 20322920	DB " 2) Caract propor
73 90B2 43617261	
73 90B6 63742070	
73 90BA 726F706F	
73 90BE 7274696F	
73 90C2 6E6E656C	
73 90C6 73	
74 90C7 0D0A	DB CR,LF
75 90C9 20332920	DB " 3) Caract conden
75 90CD 43617261	
75 90D1 63742063	
75 90D5 6F6E6465	
75 90D9 6E736573	
75 90DD 0D0A	
76 90DF 20342920	DB " 4) Standard NLQ"
76 90E3 5374616E	
76 90E7 64617264	
76 90EB 204E4C51	
76 90EF 0D0A	
77 90F1 20352920	DB " 5) Proportionnel
77 90F5 50726F70	
77 90F9 6F727469	
77 90FD 6F6E6E65	
77 9101 6C204E4C	
77 9105 510D0A	
78 9108 20362920	DB " 6) Double frappe
78 910C 446F7562	

```
78 9110 6C652066
78 9114 72617070
78 9118 650D0A
79 911B 20372920          DB   " 7) Caract italiq
79 911F 43617261
79 9123 63742069
79 9127 74616C69
79 912B 71756573
79 912F 0D0A
80 9131 20382920          DB   " 8) Caract gras",
80 9135 43617261
80 9139 63742067
80 913D 7261730D
80 9141 0A
81 9142 20392920          DB   " 9) Caract soulig
81 9146 43617261
81 914A 63742073
81 914E 6F756C69
81 9152 676E6573
81 9156 0D0A
82 9158 31302920          DB   "10) Double largeu
82 915C 446F7562
82 9160 6C65206C
82 9164 61726765
82 9168 75720D0A
82 916C FF
83
84          T:          EQU 0FFH          ;Terminateur
85          INITEV:      EQU 0BCEFH        ;INIT EVEN BLOC
86          ADDEVE:      EQU 0BCE9H        ;ADD EVEN BLOC
87          TXTOUT:      EQU 0BB5AH        ;TXT OUTPUT
```

```

88      SETCUR:    EQU  0BB75H          ;TXT SET CURSOR
89      GETCUR:    EQU  0BB78H          ;TXT GET CURSOR
90      CUREN:     EQU  0BB7BH          ;TXT CUR ENABLE
91      CURDIS:    EQU  0BB7EH          ;TXT CUR DISABL
92      LOGEXT:    EQU  0BCD1H          ;KL LOG EXT
93      PRINTCH:   EQU  0BD2BH          ;MC PRINT CHAR
94      CR:        EQU  0DH             ;Carriage Return
95      LF:        EQU  0AH             ;Line Feed
96      ;
97      ;-----
98      ; Definition de la RSX
99      ;-----
100     ;
101     DEFRSX:    EQU  $                ;Point d'entree
102 916D 011890   LD    BC, PTRTAB      ;Ptr table definition
103 9170 211490   LD    HL, BUF         ;Buffer pour LOG EXT
104 9173 CDD1BC   CALL LOGEXT          ;Definition de la RSX
105 9176 C9      RET
106     ;
107     ;-----
108     ; Traitement de la RSX
109     ; Aide ou initialisation de
110     ; l'imprimante
111     ;-----
112     ;
113     TRAITE:    EQU  $
114 9177 DD7E00   LD    A, (IX+0)
115 917A B7      OR    A
116 917B 2840   JR    Z, MODE          ;Texte explicatif

```

```

117 917D D601          SUB  1
118 917F 320390       LD   (SAVA),A          ;Sauvegarde de A
119 9182 212190       LD   HL,ONOFF         ;Table des ON/OFF
120 9185 1600         LD   D,0
121 9187 5F          LD   E,A
122 9188 19          ADD  HL,DE
123 9189 7E          LD   A,(HL)
124 918A 47          LD   B,A          ;Sauvegarde de A
125 918B B7          OR   A
126 918C 2804       JR   Z,MISEA1
127 918E AF          XOR  A
128 918F 77          LD   (HL),A
129 9190 1802       JR   SUITRAI         ;Suite traitement
130                MISEA1: EQU  $
131 9192 3C          INC  A
132 9193 77          LD   (HL),A
133                SUITRAI: EQU  $
134 9194 3A0390       LD   A,(SAVA)
135 9197 87          ADD  A,A
136 9198 87          ADD  A,A
137 9199 80          ADD  A,B
138 919A 80          ADD  A,B
139 919B 1600       LD   D,0
140 919D 5F          LD   E,A
141 919E 217590       LD   HL,ADR
142 91A1 19          ADD  HL,DE          ;HL=@ dans CONTROL
143 91A2 5E          LD   E,(HL)
144 91A3 23          INC  HL
145 91A4 56          LD   D,(HL)
146 91A5 EB          EX  DE,HL
147                BISTRAIT: EQU  $

```



```

148 91A6 7E          LD  A,(HL)
149 91A7 FEFF        CP  T           ;Terminateur
150 91A9 2806        JR  Z,FINCODE   ;Oui => Fin trait
151 91AB CD2BBD      CALL PRINTCH    ;MC SEND PRINTER
152 91AE 23          INC HL
153 91AF 18F5        JR  BISTRAIT   ;Boucle de traitement
154                  ;
155                  FINCODE: EQU $
156 91B1 3E0D        LD  A,CR
157 91B3 CD2BBD      CALL PRINTCH
158 91B6 3E0A        LD  A,LF
159 91B8 CD2BBD      CALL PRINTCH
160 91BB 180E        JR  FINTRAIT
161                  ;
162                  MODE: EQU $
163 91BD 219D90      LD  HL,AUSEC    ;@ texte explicatif
164                  BISAFF: EQU $
165 91C0 7E          LD  A,(HL)
166 91C1 FEFF        CP  0FFH
167 91C3 2806        JR  Z,FINTRAIT  ;Fin affichage
168 91C5 CD5ABB      CALL TXTOUT
169 91C8 23          INC HL         ;Prochain caractere
170 91C9 18F5        JR  BISAFF     ;Boucle d'affichage
171                  ;
172                  FINTRAIT: EQU $
173 91CB C9          RET
174                  END

```

ADR	9075 AUSEC	909D ADDEVE	BCE9 BUF	9014
BISTRAIT	91A6 BISAFF	91C0 CONTROL	902B CUREN	BB7B
CURDIS	BB7E CR	000D DEFRSX	916D EVBL	9004
FINCODE	91B1 FINTRAIT	91CB GETCUR	BB78 HE	9011
INITEV	BCEF LOGEXT	BCD1 LF	000A MI	9012
MISEA1	9192 MODE	91BD ONOFF	9021 PTRTAB	9018
PRINTCH	BD2B SAVA	9003 SE	9013 SETCUR	BB75
SUITRAI	9194 TABLE	901D T	00FF TXTOUT	BB5A
TRAITE	9177			

Assemblez-le et initialisez la RSX en tapant sous Basic :

CALL &9000

Si vous préférez utiliser un chargeur Basic, en voici le listing et les données de checksum correspondantes :

```

10 REM -----
20 REM Initialisation d'un mode d'impression pour DMP 2000
30 REM -----
40 REM
50 FOR i=&9000 TO &91CB
60   READ a$
70   a=VAL("&"+a$)
80   POKE i,a
90 NEXT i
100 END
110 REM -----
1000 DATA C3,6D,91,0,0,0,0,0,0,0,0,0,0,0,0
1010 DATA 0,0,0,0,0,0,0,1D,90,C3,77,91,49,4D,D0
1020 DATA 0,0,0,0,0,0,0,0,0,0,1B,4D,FF,1B,50
1030 DATA FF,1B,50,1,FF,1B,50,0,FF,F,FF,12,FF,1B,78,1
1040 DATA FF,1B,78,0,FF,1B,78,1,1B,70,1,FF,1B,78,0,1B
1050 DATA 70,0,FF,1B,47,FF,1B,48,FF,1B,34,FF,1B,35,FF,1B
1060 DATA 45,FF,1B,46,FF,1B,2D,1,FF,1B,2D,0,FF,1B,57,1
1070 DATA FF,1B,57,0,FF,2B,90,2E,90,31,90,35,90,39,90,3B
1080 DATA 90,3D,90,41,90,45,90,4C,90,53,90,56,90,59,90,5C
1090 DATA 90,5F,90,62,90,65,90,69,90,6D,90,71,90,20,31,29
1100 DATA 20,43,61,72,61,63,74,20,6D,69,6E,69,D,A,20,32
1110 DATA 29,20,43,61,72,61,63,74,20,70,72,6F,70,6F,72,74
1120 DATA 69,6F,6E,6E,65,6C,73,D,A,20,33,29,20,43,61,72
1130 DATA 61,63,74,20,63,6F,6E,64,65,6E,73,65,73,D,A,20
1140 DATA 34,29,20,53,74,61,6E,64,61,72,64,20,4E,4C,51,D
1150 DATA A,20,35,29,20,50,72,6F,70,6F,72,74,69,6F,6E,6E
1160 DATA 65,6C,20,4E,4C,51,D,A,20,36,29,20,44,6F,75,62
1170 DATA 6C,65,20,66,72,61,70,70,65,D,A,20,37,29,20,43
1180 DATA 61,72,61,63,74,20,69,74,61,6C,69,71,75,65,73,D
1190 DATA A,20,38,29,20,43,61,72,61,63,74,20,67,72,61,73
1200 DATA D,A,20,39,29,20,43,61,72,61,63,74,20,73,6F,75
1210 DATA 6C,69,67,6E,65,73,D,A,31,30,29,20,44,6F,75,62
1220 DATA 6C,65,20,6C,61,72,67,65,75,72,D,A,FF,1,18,90
1230 DATA 21,14,90,CD,D1,BC,C9,DD,7E,0,B7,28,40,D6,1,32
1240 DATA 3,90,21,21,90,16,0,5F,19,7E,47,B7,28,4,AF,77
1250 DATA 18,2,3C,77,3A,3,90,87,87,80,80,16,0,5F,21,75
1260 DATA 90,19,5E,23,56,EB,7E,FE,FF,28,6,CD,2B,BD,23,18
1270 DATA F5,3E,D,CD,2B,BD,3E,A,CD,2B,BD,18,E,21,9D,90
1280 DATA 7E,FE,FF,28,6,CD,5A,BB,23,18,F5,C9,0,0,0,0

```

```

C2 E1 D3 8D 63 F0 AB 79 F3 DD AB D2 C5 56 CA 57 20 6D F CA 82 D1 A7 72 C5 B7 B 6
C 8A

```

Initialisez la RSX en tapant :

CALL &9000

La RSX est maintenant opérationnelle.

Pour avoir un rappel des modes d'impression possibles, tapez :

:IMP,0

Les données suivantes s'affichent sur l'écran :

- 1) Caract mini,
- 2) Caract proportionnels,
- 3) Caract condensés,
- 4) Standard NLQ,
- 5) Proportionnel NLQ,
- 6) Double frappe,
- 7) Caract italiques,
- 8) Caract gras,
- 9) Caract soulignés,
- 10) Double largeur.

Pour activer un des modes d'impression, il suffit de préciser son numéro à la suite de l'instruction **:INP**. Par exemple, pour activer les caractères soulignés, tapez :


:IMP,9

La même instruction dévalide le mode d'impression sélectionné. Pour revenir au mode d'impression normal suite à l'initialisation de l'imprimante en mode souligné, tapez à nouveau :

:IMP,9

LE PROGRAMME EN DÉTAIL

La logique du programme obéit à l'ordinogramme suivant :



Les premières lignes du programme sont consacrées à la définition et l'initialisation de diverses variables et structures.

Remarquez en particulier :

- le bloc d'interruption ;

```

EVBL:    DB    0,0,0,0
         DB    0,0,0,0
         DB    0,0,0,0,0
  
```

— la table de déclaration de la RSX :

```

PTRTAB:  DW    TABLE
         JP    TRAITE
  
```

```
TABLE:  DB    "IM"
        DB    "P" + 80H
        DB    0
```

- la table des modes d'impression. Cette table est constituée de 10 octets. L'octet N vaut 1 lorsque le mode d'impression N est actif. Il vaut 0 lorsque le mode d'impression N est inactif :

```
ONOFF:  EQU    $
        DB    0,0,0,0,0,0,0,0,0,0
```

- la table des codes de contrôle. Cette table contient tous les codes de contrôle à envoyer à l'imprimante pour initialiser ses modes d'impression :

```
CONTROL: EQU    $
        DB    27,"M",T,27,"P",T
```

Remarquez dans cette table la présence de codes terminateurs (T = 255) qui permettent au programme de terminer chaque séquence d'initialisation.

- la table d'indirection dans CONTROL qui donne l'adresse de chaque séquence d'initialisation :

```
ADR:    EQU    $
        DW    CONTROL
        ...
        DW    CONTROL + 70
```

- la table d'aide qui contient la liste des modes d'impression possibles :

```
AUSEC:  EQU    $
        DB    " 1) Caract mini",CR,LF
        ...
        DB    "10) Double largeur",CR,LF
```

Le court programme situé à l'étiquette DEFERSX définit la RSX IMP. Pour cela, il fait appel à la macro LOGEXT du firmware :

```
DEFERSX: EQU    $
        LD    BC,PTRTAB
        LD    HL,BUFF
        CALL LOGEXT
        RET
```

Lorsque vous activez la RSX IMP en tapant : IMP, <Code> sous Basic, l'interpréteur donne le contrôle au programme situé à l'étiquette TRAITE.

Ce programme compare la valeur du paramètre qui est passée à la RSX à zéro :

```
TRAITE: EQU    $
        LD    A,(IX + 0)
        OR    A
```

Si ce paramètre est nul, les divers modes d'impression possibles sont affichés sur l'écran :

```
JR      Z,MODE
```

L'affichage des modes d'impression possibles se fait à l'intérieur d'une boucle qui prend fin lorsque le caractère terminateur **OFFH** est rencontré :

```
MODE:   EQU    $
        LD     HL,AUSEC

BISAFF: EQU    $
        LD     A,(HL)
        CP     OFFH
        JR     Z,FINTRAIT
        CALL  TXTOUT
        INC   HL
        JR     BISAFF
```

Lorsque le paramètre passé à la RSX n'est pas nul, il est décrémenté de un et sauvegardé dans la variable **SAVA** :

```
SUB    1
LD     (SAVA),A
```

La table des modes actifs est consultée. L'octet correspondant est stocké dans le registre **B** :

```
LD     HL,ONOFF
LD     D,0
LD     E,A
ADD    HL,DE
LD     A,(HL)
LD     B,A
```

Si le code correspondant au mode sélectionné est à un, il est mis à zéro. S'il est à zéro, il est mis à un :

```
OR     A
JR     Z,MISEA1
XOR    A
LD     (HL),A
JR     SUI TRAI

MISEA1: EQU    $
        INC   A
        LD   (HL),A

SUI TRAI: EQU    $
```

L'adresse des codes à envoyer à l'imprimante est trouvée par la formule suivante :

$$\text{Adresse} = \text{ADR} + 4 * \langle \text{Mode demandé} \rangle + 2 * \langle \text{Mode actif} \rangle$$

où $\langle \text{Mode demandé} \rangle$ est le paramètre passé à la RSX, et $\langle \text{Mode actif} \rangle$ vaut 1 si le mode demandé est déjà actif et 0 dans le cas contraire.

Ce qui se traduit par les instructions Assembleur suivantes :

```
LD    A,(SAVA)
ADD   A,A
ADD   A,A
ADD   A,B
ADD   A,B
LD    D,0
LD    E,A
LD    HL,ADR
ADD   HL,DE
```

A ce niveau, le registre HL pointe dans la table ADR sur l'adresse de la séquence à envoyer à l'imprimante. Cette adresse est chargée dans le registre HL avec les instructions suivantes :

```
LD    E,(HL)
INC   HL
LD    D,(HL)
EX   DE,HL
```

Le registre HL contient l'adresse du premier code à envoyer à l'imprimante. Les codes de définition de mode sont envoyés à l'intérieur d'une boucle qui prend fin lorsque le caractère terminateur **OFFH** est rencontré :

```
BISTRAIT: EQU    $
          LD     A,(HL)
          CP     T
          JR     Z,FINCODE
          CALL   PRINTCH
          INC   HL
          JR     BISTRAIT
```

Lorsque les codes de formatage sont envoyés, le programme termine la séquence avec les codes CR, LF, ce qui provoque un passage à la ligne :

```
FINCODE: EQU    $
          LD     A,CR
          CALL   PRINTCH
          LD     A,LF
          CALL   PRINTCH
```